```
;{CR10X}

;{CR10X}
;  Non-acid Tussock     Edited: 16Aug99 JL&SH; 13Jun05 Added year to outpput.  JL
; 19Oct07 added a subroutnie to use Steinhart-Hart equation for reference temperature.  JL
; 6Oct11 Took out the hour on final storage for daily
; 10Aug14  Added Raido control since switched radio to a Freeway.
; Added program signature and reference tempertures -both subroutine and P11 values  JimL

; NOTES: Soil temps in the LTER NonAcidic Tussock site treatment plots using a AM416 multiplexer.
; 2 strings per treatment with sur, 10, 20, 40 in
; CT, NP, GH, GHF in block 3.
;
;Input Channel Usage: 1 Reference temp, 2SE Air Temp CT, 3SE Air temp GH,
;                4SE RH CT, 5SE RH GH, 5DE Mult AM416, 6DE Mult AM416
;Note: Using 1 SE for ref. temp.
;Excitation Channel Usage: 3 Reference Temp
;Control Port Usage:1 res AM416, 2 Clock AM416
;Pulse Input Channel Usage:
;
;Output Array Definitions:
;Table 1:
;
;Hourly:  ID, Julian, Hour, CT Air Temp, GH air Temp, CT RH, GH RH
;
;Daily:   ID, Julian, CT Air MAX, Hour-min of MAX, GH Air MAX, Hour-min of MAX,
;         CT Air MIN Hour-min of MIN, GH Air MIN, Hour-min of MIN
;
;Table 2:
;
;3Hourly:   ID, Julian, Hour,
;         NP1 Sur, NP1 10cm, NP1 20cm, NP1 40cm,
;         NP2 Sur, NP2 10cm, NP2 20cm, NP2 40cm,
;         CT1 Sur, CT1 10cm, CT1 20cm, CT1 40cm,
;         CT2 Sur, CT2 10cm, CT2 20cm, CT2 40cm,
;         GH1 Sur, GH1 10cm, GH1 20cm, GH1 40cm,
;         GH2 Sur, GH2 10cm, GH2 20cm, GH2 40cm,
;         GHF1 Sur, GHF1 10cm, GHF1 20cm, GHF1 40cm,
;         GHF2 Sur, GHF2 10cm, GHF2 20cm, GHF2 40cm,
;
; WIRING NOTES:
; Temp/RH: Temp -Black; RH - Brown; Ground - Green & Clear; 12 volt - Red
;

*Table 1 Program
  01: 60       Execution Interval (seconds)

;Measure CS500 Temperature
```

```
1:  Volts (SE) (P1)
 1: 2        Reps
 2: 5        2500 mV Slow Range
 3: 2        SE Channel
 4: 2        Loc [ CT_3M_Air ]
 5: .1       Mult
 6: -40      Offset

;Measure CS500 Relative Humidity

2:  Volts (SE) (P1)
 1: 2        Reps
 2: 5        2500 mV Slow Range
 3: 4        SE Channel
 4: 4        Loc [ CT_3M_RH  ]
 5: .1       Mult
 6: 0.0      Offset

;Limit the maximum humidity to 100%

3:  IF (X<=>F) (P89)
 1: 4        X Loc [ CT_3M_RH  ]
 2: 3        >=
 3: 100      F
 4: 30       Then Do

    4:  Z=F (P30)
     1: 100      F
     2: 00       Exponent of 10
     3: 4        Z Loc [ CT_3M_RH  ]

5:  End (P95)

6:  IF (X<=>F) (P89)
 1: 5        X Loc [ GH_RH     ]
 2: 3        >=
 3: 100      F
 4: 30       Then Do

    7:  Z=F (P30)
     1: 100      F
     2: 00       Exponent of 10
     3: 5        Z Loc [ GH_RH     ]

8:  End (P95)

9:  If time is (P92)
 1: 0000     Minutes (Seconds --) into a
 2: 60       Interval (same units as above)
 3: 10       Set Output Flag High

10:  Real Time (P77)
 1: 1220     Year,Day,Hour/Minute (midnight = 2400)
```

```
11:  Average (P71)
 1: 4        Reps
 2: 2        Loc [ CT_3M_Air ]

12:  If time is (P92)
 1: 0000     Minutes (Seconds --) into a
 2: 1440     Interval (same units as above)
 3: 10       Set Output Flag High

13:  Real Time (P77)
 1: 1200     Year,Day (midnight = 2400)

14:  Maximize (P73)
 1: 2        Reps
 2: 10       Value with Hr-Min
 3: 2        Loc [ CT_3M_Air ]

15:  Minimize (P74)
 1: 2        Reps
 2: 10       Value with Hr-Min
 3: 2        Loc [ CT_3M_Air ]

16:  Signature (P19)
 1: 51       Loc [ ProgSig   ]

17:  Sample (P70)
 1: 1        Reps
 2: 51       Loc [ ProgSig   ]

18:  Serial Out (P96)
 1: 71       SM192/SM716/CSM1

; Radio on control

19:  Batt Voltage (P10)
 1: 38       Loc [ Battery   ]

; Setup the time of day to begin and end when the radio will turn on
; Here it will start at 10:00 and end at 18:00
20:  Z=F x 10^n (P30)
 1: 600      F
 2: 00       n, Exponent of 10
 3: 48       Z Loc [ PdStart   ]

21:  Z=F x 10^n (P30)
 1: 1080     F
 2: 00       n, Exponent of 10
 3: 49       Z Loc [ PdEnd     ]

22:  Time (P18)
 1: 1        Minutes into current day (maximum 1440)
 2: 0000     Mod/By
 3: 47       Loc [ Now_min   ]
```

```
23:  Do (P86)
 1: 21       Set Flag 1 Low

24:  If (X<=>Y) (P88)
 1: 47       X Loc [ Now_min   ]
 2: 3        >=
 3: 48       Y Loc [ PdStart   ]
 4: 30       Then Do

    25:  If (X<=>Y) (P88)
     1: 47       X Loc [ Now_min   ]
     2: 4        <
     3: 49       Y Loc [ PdEnd     ]
     4: 30       Then Do

        26:  Z=X-Y (P35)
         1: 47       X Loc [ Now_min   ]
         2: 48       Y Loc [ PdStart   ]
         3: 50       Z Loc [ Min       ]

; Calculate the minutes in current hour

        27:  Z=X MOD F (P46)
         1: 50       X Loc [ Min       ]
         2: 60       F
         3: 50       Z Loc [ Min       ]

;From 30 to 60 minutes set flag 1 high

        28:  If (X<=>F) (P89)
         1: 50       X Loc [ Min       ]
         2: 3        >=
         3: 30       F
         4: 11       Set Flag 1 High

        29:  If (X<=>F) (P89)
         1: 38       X Loc [ Battery   ]
         2: 4        <
         3: 12.2     F
         4: 21       Set Flag 1 Low

    30:  End (P95)

31:  End (P95)

;Turn Radio on

32:  If Flag/Port (P91)
 1: 11       Do if Flag 1 is High
 2: 30       Then Do

    33:  Do (P86)
     1: 43       Set Port 3 High
```

```
34:  Else (P94)

    35:  Do (P86)
     1: 53       Set Port 3 Low

36:  End (P95)

*Table 2 Program
  02: 900       Execution Interval (seconds)

; Call subroutine to read reference temperature
1:  Do (P86)
 1: 1        Call Subroutine 1

2:  Do (P86) ; (ACTIVATES MULTIPLEXER) with control port 1
 1: 41       Set Port 1 High

3:  Beginning of Loop (P87); BEGIN MEASUREMENT LOOP
 1: 0        Delay
 2: 16       Loop Count

    4:  Do (P86) ; CLOCK PLUS with control port 2
     1: 72       Pulse Port 2

    5:  Thermocouple Temp (DIFF) (P14); MEASURE SETS 1-16
     1: 1        Reps
     2: 1        2.5 mV Slow Range
     3: 5        DIFF Channel
     4: 1        Type T (Copper-Constantan)
     5: 1        Ref Temp Loc [ REF_TEMP  ]
     6: 6     -- Loc [ NP1Sur    ]
     7: 1        Mult
     8: 0        Offset

    6:  Thermocouple Temp (DIFF) (P14); MEASURE SETS 17-32
     1: 1        Reps
     2: 1        2.5 mV Slow Range
     3: 6        DIFF Channel
     4: 1        Type T (Copper-Constantan)
     5: 1        Ref Temp Loc [ REF_TEMP  ]
     6: 22    -- Loc [ GH1sur    ]
     7: 1        Mult
     8: 0        Offset

7:  End (P95); ENDS LOOP

8:  Do (P86)   ;Deactivates Multiplexer
 1: 51       Set Port 1 Low

9:  Batt Voltage (P10)
 1: 38       Loc [ Battery   ]
```

```
10:  Temp (107) (P11)
 1: 1        Reps
 2: 1        SE Channel
 3: 3        Excite all reps w/E3
 4: 53       Loc [ Ref_107   ]
 5: 1.0      Multiplier
 6: 0.0      Offset

11:  If time is (P92)
 1: 0000     Minutes (Seconds --) into a
 2: 180      Interval (same units as above)
 3: 10       Set Output Flag High

12:  Set Active Storage Area (P80)
 1: 1        Final Storage Area 1
 2: 210      Array ID

13:  Real Time (P77)
 1: 1220     Year,Day,Hour/Minute (midnight = 2400)

14:  Average (P71)
 1: 34       Reps
 2: 6        Loc [ NP1Sur    ]

15:  Average (P71)
 1: 1        Reps
 2: 1        Loc [ REF_TEMP  ]

16:  Average (P71)
 1: 1        Reps
 2: 53       Loc [ Ref_107   ]

17:  Serial Out (P96)
 1: 71       SM192/SM716/CSM1

*Table 3 Subroutines
1:  Beginning of Subroutine (P85)
 1: 1        Subroutine 1
; Steinhart Hart equation for temperature
; Read reference temperature with a bridge statement and then
; convert the bridge voltage to resistance
; using Rs=Rf(Vx/Vs-1)-Rb where bridge resistors Rf and Rb
; are 1 and 249 K ohms and Vs is the voltage read
; and Vx is the excitation voltage.
; The Steinhart Haart equation is
; 1/T = A + B(ln Rs) + C(ln Rs)^3
; where coefficients are: A  8.4634e-04, B 2.0627e-04 and C 8.6958e-08
; for the Fenwall UUT51J1 Thermistor for Campbell's 107 probe
; Note that Campbell changed the thermistor to a Belatherm 100K6A
; thermistor probably some time around 2000. This will change the coefficients.
; From Technical Memorandum by Gary Clow USGS Menlo Park, CA
; Note that the polynomial instruction, 55, is used to apply the
; Steinhart and Hart equation. Instruction 55 does not allow entering
; the coefficients with scientific notation. In order to use this instruction
```

```
; with as much resolution as possible, the ln resistance term is pre scaled by 10^-3.
; This allows the first order coefficient (B) to be multiplied by 10^3 and
; the 3rd order coefficient (C) to be multiplied by 10^9.
; Note program adapted from 109 manual

    ; Read the thermistor with a half bridge instruction
    ; use excitation mv 2000 for CR10 and 4000 for 21x
    2:  AC Half Bridge (P5)
     1: 1        Reps
     2: 22       7.5 mV 60 Hz Rejection Range
     3: 1        SE Channel
     4: 3        Excite all reps w/Exchan 3
     5: 2000     mV Excitation
     6: 39       Loc [ V_Vx      ]
     7: 1.0      Multiplier
     8: 0.0      Offset

    ; Change to Vx/Vs
    3:  Z=1/X (P42)
     1: 39       X Loc [ V_Vx      ]
     2: 40       Z Loc [ Vx_V      ]

    ; Rs in K ohms (Rf*(Vx/Vs-1))-Rb where Rf = 1 and RB = 249
    4:  Z=X+F (P34)
     1: 40       X Loc [ Vx_V      ]
     2: -250     F
     3: 41       Z Loc [ RthermK   ]

    ; Convert K ohms to ohms
    5:  Z=X*F (P37)
     1: 41       X Loc [ RthermK   ]
     2: 1000     F
     3: 42       Z Loc [ Rtherm    ]

    ; ln(Rs)
    6:  Z=LN(X) (P40)
     1: 42       X Loc [ Rtherm    ]
     2: 43       Z Loc [ lnRt      ]

    ; Scale Rs by 10^-3
    7:  Z=X*F (P37)
     1: 43       X Loc [ lnRt      ]
     2: .001     F
     3: 44       Z Loc [ Scal_lnRt ]

    ; Apply equation coefficients
```

```
   8:  Polynomial (P55)
    1: 1        Reps
    2: 44       X Loc [ Scal_lnRt ]
    3: 45       F(X) Loc [ I_Tk      ]
    4: 0        C0
    5: .20880   C1
    6: 0        C2
    7: 80.592   C3
    8: 0        C4
    9: 0        C5

   9:  Z=F x 10^n (P30)
    1: 8.2711   F
    2: -4       n, Exponent of 10
    3: 52       Z Loc [ co        ]

   10:  Z=X+Y (P33)
    1: 45       X Loc [ I_Tk      ]
    2: 52       Y Loc [ co        ]
    3: 45       Z Loc [ I_Tk      ]

   ; 1/((A+B*ln(Rs)) + C*(ln(Rs))^3) = Kelvin
   11:  Z=1/X (P42)
    1: 45       X Loc [ I_Tk      ]
    2: 46       Z Loc [ Tk        ]

   ; Convert to Celsius
   12:  Z=X+F (P34)
    1: 46       X Loc [ Tk        ]
    2: -273.15  F
    3: 1        Z Loc [ REF_TEMP  ]

;end sub 1
13:  End (P95)

End Program
```

Addr	Name	Flags	# Reads	# Writes	Blocks
1	[REF_TEMP]	RWM-	3	1	----- ------ ---
2	[CT_3M_Air]	RWM-	3	1	Start ------ ---
3	[GH_Air]	RWM-	3	1	----- ------ End
4	[CT_3M_RH]	RWM-	2	2	Start ------ ---
5	[GH_RH]	RWM-	2	2	----- ------ End
6	[NP1Sur]	RWM-	1	1	----- ------ ---
7	[NP1_10cm]	R-M-	1	0	----- ------ ---
8	[NP1_20cm]	R-M-	1	0	----- ------ ---
9	[NP1_40cm]	R-M-	1	0	----- ------ ---
10	[NP2sur]	R-M-	1	0	----- ------ ---
11	[NP2_10cm]	R-M-	1	0	----- ------ ---
12	[NP2_20cm]	R-M-	1	0	----- ------ ---
13	[NP2_40cm]	R-M-	1	0	----- ------ ---
14	[CT1sur]	R-M-	1	0	----- ------ ---
15	[CT1_10cm]	R-M-	1	0	----- ------ ---
16	[CT1_20cm]	R-M-	1	0	----- ------ ---
17	[CT1_40cm]	R-M-	1	0	----- ------ ---
18	[CT2sur]	R-M-	1	0	----- ------ ---
19	[CT2_10cm]	R-M-	1	0	----- ------ ---
20	[CT2_20cm]	R-M-	1	0	----- ------ ---
21	[CT2_40cm]	R-M-	1	0	----- ------ ---
22	[GH1sur]	RWM-	1	1	----- ------ ---
23	[GH1_10cm]	R-M-	1	0	----- ------ ---
24	[GH1_20cm]	R-M-	1	0	----- ------ ---
25	[GH1_40cm]	R-M-	1	0	----- ------ ---
26	[GH2sur]	R-M-	1	0	----- ------ ---
27	[GH2_10cm]	R-M-	1	0	----- ------ ---
28	[GH2_20cm]	R-M-	1	0	----- ------ ---
29	[GH2_40cm]	R-M-	1	0	----- ------ ---
30	[GHF1sur]	R-M-	1	0	----- ------ ---
31	[GHF1_10cm]	R-M-	1	0	----- ------ ---
32	[GHF1_20cm]	R-M-	1	0	----- ------ ---
33	[GHF1_40cm]	R-M-	1	0	----- ------ ---
34	[GHF2sur]	R-M-	1	0	----- ------ ---
35	[GHF2_10cm]	R-M-	1	0	----- ------ ---
36	[GHF2_20cm]	R-M-	1	0	----- ------ ---
37	[GHF2_40cm]	R-M-	1	0	----- ------ ---
38	[Battery]	RW--	2	2	----- ------ ---
39	[V_Vx]	RW--	1	1	----- ------ ---
40	[Vx_V]	RW--	1	1	----- ------ ---
41	[RthermK]	RW--	1	1	----- ------ ---
42	[Rtherm]	RW--	1	1	----- ------ ---
43	[lnRt]	RW--	1	1	----- ------ ---
44	[Scal_lnRt]	RW--	1	1	----- ------ ---
45	[I_Tk]	RW--	2	2	----- ------ ---
46	[Tk]	RW--	1	1	----- ------ ---
47	[Now_min]	RW--	3	1	----- ------ ---
48	[PdStart]	RW--	2	1	----- ------ ---
49	[PdEnd]	RW--	1	1	----- ------ ---
50	[Min]	RW--	2	2	----- ------ ---
51	[ProgSig]	RW--	1	1	----- ------ ---
52	[co]	RW--	1	1	----- ------ ---

53 [Ref_107] RW-- 1 1 ----- ------ ---